



WOLFRAM BLOG



Hidden Figures: Modern Approaches to Orbit and Reentry Calculations

February 24, 2017

[Jeffrey Bryant](#), Research Programmer, Wolfram|Alpha Scientific Content

[Paco Jain](#), Research Programmer, Wolfram|Alpha Scientific Content

[Michael Trott](#), Chief Scientist



The movie [Hidden Figures](#) was released in theaters recently and has been getting good reviews. It also deals with an important time in US history, touching on a number of topics, including civil rights and the Space Race. The movie details the hidden story of [Katherine Johnson](#) and her coworkers ([Dorothy Vaughan](#) and [Mary Jackson](#)) at NASA during the Mercury missions and the United States' early explorations into manned space flight. The movie focuses heavily on the dramatic civil rights struggle of African American women in NASA at the time, and these struggles are set against the number-crunching ability of Johnson and her coworkers. Computers were in their early days at this time, so Johnson and her team's ability to perform complicated navigational orbital mechanics problems without the use of a computer provided an important sanity check against the early computer results.

```
In[1]:= Row[{Show[Hidden Figures \(movie\) [image], ImageSize -> 101], " ",
Show[Katherine Johnson-like curve \(popular curve\) [image], Axes -> False,
Background -> LightBlue, ImageSize -> 120]]]
```




I will touch on two aspects of her scientific work that were mentioned in the film: orbit calculations and reentry calculations. For the orbit calculation, I will first exactly follow what Johnson did and then compare with a more modern, direct approach utilizing an array of tools made available with the [Wolfram Language](#). Where the movie mentions the solving of differential equations using [Euler's method](#), I will compare this method with more modern ones in an important problem of rocketry: computing a reentry trajectory from the rocket equation and drag terms (derived using atmospheric model data obtained directly from within the Wolfram Language).

The movie doesn't focus much on the math details of the types of problems Johnson and her team dealt with, but for the purposes of this blog, I hope to provide at least a flavor of the approaches one might have used in Johnson's day compared to the present.

Placing a Satellite over a Selected Position

One of the earliest papers that Johnson coauthored, "[Determination of Azimuth Angle at Burnout for Placing a Satellite over a Selected Earth Position](#)," deals with the problem of making sure that a satellite can be placed over a specific Earth location



Constants and Initial Processing

In the paper, Johnson defines a number of constants and input parameters needed to solve the problem at hand. One detail to explain is the term “burnout,” which refers to the shutoff of the rocket engine. After burnout, orbital parameters are essentially “frozen,” and the spacecraft moves solely under the Earth’s gravity (as determined, of course, through Newton’s laws). In this section, I follow the paper’s unit conventions as closely as possible.



```

pericenter = 210000000; (* p is the semilatus rectum of the orbit ellipse *)
a = 22081775.57; (* semimajor axis of orbit in feet *)
t01 = 5.842; (* t[01] is the time from perigee for 01 in min,
where 01 is angle in orbit plane between perigee and burnout *)
g0 = 115991.595; (* acceleration due to gravity ft/min^2 *)
R = 2.090226 * 10^7; (* Earth radius in feet *)
(* launch coordinates *)
phi1 = 28.50; (* launch latitude *)
lambda1 = 279.45; (* launch longitude *)
phi2 = 34.00; (* intended pass over latitude *)
lambda2 = 241.00; (* intended pass over longitude *)
n = 3; (* number of orbits *)
omegaCapitale = 0.25068 (* angular velocity of Earth in degrees/min *);

```

For convenience, some functions are defined to deal with angles in degrees instead of radians. This allows for smoothly handling time in angle calculations:

```

In[17]:= SinDegree[x_] := Sin[x Degree]
CosDegree[x_] := Cos[x Degree]
TanDegree[x_] := Tan[x Degree]
SecDegree[x_] := Sec[x Degree]

ArcSinDegree[x_] := ArcSin[x]/Degree
ArcCosDegree[x_] := ArcCos[x]/Degree
ArcTanDegree[x_] := ArcTan[x]/Degree

In[24]:= ArcTanDegree[x_, y_] := ArcTan[x, y]/Degree

```

Johnson goes on to describe several other derived parameters, though it's interesting to note that she sometimes adopted values for these rather than using the values returned by her formulas. Her adopted values were often close to the values obtained by the formulas. For simplicity, the values from the formulas are used here.

Semilatus rectum of the orbit ellipse:

```

In[25]:= p = r1 * (v1/vc)^2 CosDegree[y1];

```

Angle in orbit plane between perigee and burnout point:

```

In[26]:= theta1 = ArcTanDegree[TanDegree[y1] ((p/r1)/(p/r1 - 1))];

```

Orbit eccentricity:

```

In[27]:= e = (1/CosDegree[theta1]) (p/r1 - 1);

```

Orbit period:

```

In[28]:= T = 2 Pi Sqrt[R/g0] Sqrt[(a/R)^3];

```

Eccentric anomaly:



most often, the extreme place over an equivalent position with latitude ϕ_2 the same as that of the selected position but with longitude λ_{2e} displaced eastward from λ_2 by an amount sufficient to compensate for the rotation of the Earth during the n complete orbits, that is, by the polar hour angle $n \omega_E T$. The longitude of this equivalent position is thus given by the relation":

```
In[30]:=  $\lambda_{2e} = \lambda_2 + n \omega_{\text{Capitale}} T;$ 
```


Time from perigee for angle θ :

```
In[31]:=  $t[\theta] := T / (2 \text{ Pi}) (\text{Eanomaly}[\theta \text{ Degree}] - e \text{ Sin}[\text{Eanomaly}[\theta \text{ Degree}]])$ 
```

Computation

Part of the final solution is to determine values for intermediate parameters $\delta\lambda_{1-2e}$ and θ_{2e} . This can be done in a couple of ways. First, I can use [ContourPlot](#) to obtain a graphical solution via equations 19 and 20 from the paper:

```
In[32]:= Clear[ $\Delta\lambda_{1\text{minus}2e}$ ,  $\theta_{2e}$ ];
ContourPlot[
  Evaluate[{ $\Delta\lambda_{1\text{minus}2e} == \lambda_{2e} - \lambda_1 + \omega_{\text{Capitale}} (t[\theta_{2e}] - t[\theta_1])$ ,
    CosDegree[ $\theta_{2e} - \theta_1$ ] == SinDegree[ $\phi_2$ ] SinDegree[ $\phi_1$ ] +
    CosDegree[ $\phi_2$ ] CosDegree[ $\phi_1$ ] CosDegree[ $\Delta\lambda_{1\text{minus}2e}$ ] }],
  { $\Delta\lambda_{1\text{minus}2e}$ , 0, 60}, { $\theta_{2e}$ , 0, 60},
  Prolog -> {Red, PointSize[Large], Point[{31.948028324349036`, 51.7127788640602`]}}]
```



[FindRoot](#) can be used to find the solutions numerically:

```
In[34]:= Clear[ $\Delta\lambda_{1\text{minus}2e}$ ,  $\theta_{2e}$ ];
FindRoot[
  Evaluate[{ $\Delta\lambda_{1\text{minus}2e} == \lambda_{2e} - \lambda_1 + \omega_{\text{Capitale}} (t[\theta_{2e}] - t[\theta_1])$ ,
    CosDegree[ $\theta_{2e} - \theta_1$ ] == SinDegree[ $\phi_2$ ] SinDegree[ $\phi_1$ ] +
    CosDegree[ $\phi_2$ ] CosDegree[ $\phi_1$ ] CosDegree[ $\Delta\lambda_{1\text{minus}2e}$ ] }],
  {{ $\Delta\lambda_{1\text{minus}2e}$ , 30}, { $\theta_{2e}$ , 55}}]
```




iterative method. Because the base computations are for a spherical Earth, corrections for oblateness are included in her method:

```
In[36]:= Δω = 0;
Δφ2 = 0;
ΔΩ = 0;
Δλ2 = 0;
i0 = 2;

Table[
  λ2e = λ2 - Δλ2 + n ωCapitale T;
  Δλ1minus2e = λ2e - λ1 + ωCapitale *
    If[iter == 1, T / 360 (λ2e - λ1) + t{θ1}, (tOfθ2e - t{θ1})];
  θ2e = ArcCosDegree[SinDegree[φ2 - Δφ2] SinDegree[φ1] +
    CosDegree[φ2 - Δφ2] CosDegree[φ1] CosDegree[Δλ1minus2e]] + θ1;
  ψ1 = ArcSinDegree[(SinDegree[Δλ1minus2e] CosDegree[φ2 - Δφ2]) / SinDegree[θ2e - θ1]];
  i =
    ArcCosDegree[
      Piecewise[{{CosDegree[φ1] SinDegree[ψ1], 0 < ψ1 < 180},
        {-CosDegree[φ1] SinDegree[ψ1], 180 ≤ ψ1 < 360}}];
  ω = ArcSinDegree[SinDegree[φ1] / SinDegree[i]] - θ1;
  (* after the first iteration,
    correct λ2 and φ2 for oblateness and
    then keep this correction | page 18, 19 *)
  If[iter == i0, Δω = 3.4722*^-3 (R / p)^2 (R / a)^(3 / 2) (5 CosDegree[i]^2 - 1) (n T + t{θ2e} - t{θ1})];
  If[iter == i0, Δφ2 = Δω SinDegree[i] CosDegree[ω + θ2e] / CosDegree[φ2]];
  If[iter == i0, ΔΩ = -6.9444*^-3 (R / p)^2 (R / a)^(3 / 2) CosDegree[i] (n T + t{θ2e} - t{θ1})];
  If[iter == i0, Δλ2 = Δω CosDegree[i] SecDegree[ω + θ2e]^2 /
    (1 + CosDegree[i]^2 TanDegree[ω + θ2e]^2) + ΔΩ];
  ΔλN1 = ArcTanDegree[SinDegree[φ1] TanDegree[ψ1]];
  λNref = λ1 - ΔλN1;
  tOfθ2e = t{θ2e};
  {θ2e, tOfθ2e}, {iter, 1, 8, 1}];
```

Graphing the value of θ_{2e} for the various iterations shows a quick convergence:

```
In[42]:= ListPlot[%[[All, 2]], PlotRange → {10, 15}, Filling → Axis,
  AxesLabel → {"iteration", "θ2e"}, ImageSize → 360]
```





$\omega \rightarrow 34.5578, \Delta\omega \rightarrow 1.96276, \Delta\Omega \rightarrow -1.33792, i \rightarrow 34.0355, \Delta\phi_2 \rightarrow 0.0841712\}$

I can convert the method in a `FindRoot` command as follows (this takes the oblateness effects into account in a fully self-consistent manner and calculates values for all nine variables involved in the equations):

```
In[44]:= fr := FindRoot[Evaluate[
  { $\Delta\lambda_{1\text{minus}2e} == (\lambda_2 - \Delta\lambda_2 + n \omega \text{Capitale } T) - \lambda_1 + \omega \text{Capitale} * (t[\theta_{2e}] - t[\theta_1])$ ,
     $\text{CosDegree}[\theta_{2e} - \theta_1] == \text{SinDegree}[\phi_2 - \Delta\phi_2] \text{SinDegree}[\phi_1] +$ 
       $\text{CosDegree}[\phi_2 - \Delta\phi_2] \text{CosDegree}[\phi_1] \text{CosDegree}[\Delta\lambda_{1\text{minus}2e}]$ ,
     $\text{SinDegree}[\psi_1] == (\text{SinDegree}[\Delta\lambda_{1\text{minus}2e}] \text{CosDegree}[\phi_2 - \Delta\phi_2]) / \text{SinDegree}[\theta_{2e} - \theta_1]$ ,
     $\text{CosDegree}[i] == \text{CosDegree}[\phi_1] \text{SinDegree}[\psi_1]$ ,
     $\text{SinDegree}[\omega + \theta_1] == \text{SinDegree}[\phi_1] / \text{SinDegree}[i]$ ,
     $\Delta\omega == 3.4722 * \text{a}^{-3} (R/p)^2 (R/a)^{3/2} (5 \text{CosDegree}[i]^2 - 1) (n T + t[\theta_{2e}] - t[\theta_1])$ ,
     $\Delta\phi_2 == \Delta\omega \text{SinDegree}[i] \text{CosDegree}[\omega + \theta_{2e}] / \text{CosDegree}[\phi_2 - \Delta\phi_2]$ ,
     $\Delta\Omega == -6.9444 * \text{a}^{-3} (R/p)^2 (R/a)^{3/2} \text{CosDegree}[i] (n T + t[\theta_{2e}] - t[\theta_1])$ ,
     $\Delta\lambda_2 ==$ 
       $\Delta\omega \text{CosDegree}[i] \text{SecDegree}[\omega + \theta_{2e}]^2 / (1 + \text{CosDegree}[i]^2 \text{TanDegree}[\omega + \theta_{2e}]^2) + \Delta\Omega$ ],
  {{ $\Delta\lambda_{1\text{minus}2e}$ , 31}, { $\Delta\lambda_2$ , 1}, { $\theta_{2e}$ , 51}, { $\psi_1$ , 70}, { $\omega$ , 34}, { $\Delta\omega$ , 2}, { $\Delta\Omega$ , -1.3}, { $i$ , 34}, { $\Delta\phi_2$ , 0.1}}]
```

```
In[45]:= Clear[ $\Delta\lambda_{1\text{minus}2e}$ ,  $\Delta\lambda_2$ ,  $\theta_{2e}$ ,  $\psi_1$ ,  $\omega$ ,  $\Delta\omega$ ,  $\Delta\Omega$ ,  $i$ ,  $\Delta\phi_2$ ];
wlsol = fr
```

```
Out[46]:= { $\Delta\lambda_{1\text{minus}2e} \rightarrow 31.062$ ,  $\Delta\lambda_2 \rightarrow 1.02664$ ,  $\theta_{2e} \rightarrow 50.9332$ ,  $\psi_1 \rightarrow 70.5964$ ,
   $\omega \rightarrow 34.61$ ,  $\Delta\omega \rightarrow 1.96527$ ,  $\Delta\Omega \rightarrow -1.33778$ ,  $i \rightarrow 34.0139$ ,  $\Delta\phi_2 \rightarrow 0.102921$ }
```

Interestingly, even the iterative root-finding steps of this more complicated system converge quite quickly:

```
In[47]:= {OwnValues[fr] /. HoldPattern[FindRoot[args_]] => Reap[FindRoot[args, StepMonitor => Sow[
  { $\Delta\lambda_{1\text{minus}2e}$ ,  $\Delta\lambda_2$ ,  $\theta_{2e}$ ,  $\psi_1$ ,  $\omega$ ,  $\Delta\omega$ ,  $\Delta\Omega$ ,  $i$ ,  $\Delta\phi_2$ }]}][[1, 2]][[2]]

Out[47]:= {{{{31.0619, 1.02671, 50.9332, 70.5861, 34.6025, 1.96521, -1.33777, 34.0149, 0.102885},
  {31.062, 1.02664, 50.9332, 70.5964, 34.61, 1.96527, -1.33778, 34.0139, 0.102922},
  {31.062, 1.02664, 50.9332, 70.5964, 34.61, 1.96527, -1.33778, 34.0139, 0.102921},
  {31.062, 1.02664, 50.9332, 70.5964, 34.61, 1.96527, -1.33778, 34.0139, 0.102921}}}}
```

Plotting

With the orbital parameters determined, it is desirable to visualize the solution. First, some critical parameters from the previous solutions need to be extracted:

```
In[48]:= { $\theta_{2e}$ ,  $\omega$ ,  $i$ } = {" $\theta_{2e}$ ", " $\omega$ ", " $i$ "} /. ksol;
```

Next, the latitude and longitude of the satellite as a function of azimuth angle need to be derived:

```
In[49]:=  $\Delta\lambda\text{Ns}[\phi_] := \text{ArcTanDegree}[\text{CosDegree}[\omega + \phi], \text{CosDegree}[-i] \text{SinDegree}[\omega + \phi]]$ 
```

```
In[50]:=  $\lambda\text{N}[\phi_] := \lambda\text{Nref} - \omega \text{Capitale} (t[\phi] - t[\theta_1]) + \Delta\lambda\text{Ns}[\phi]$ 
```

ϕs and λs are the latitudes and longitudes as a function of θs :




```
In[53]:= satpts[n_] := Table[{φs[θs], λs[θs, n]}, {θs, 0, 360, .01}];
```


Johnson's paper presents a sketch of the orbital solution including markers showing the burnout, selected and equivalent positions. It's easy to reproduce a similar plain diagram here:

```
In[54]:= points = {Table[{λs[θs, 0], φs[θs]}, {θs, θ1, 180, .1}],
  Table[{λs[θs, 1], φs[θs]}, {θs, 0, 360, .1}],
  Table[{λs[θs, 2], φs[θs]}, {θs, 0, 360, .1}],
  Table[{λs[θs, 3], φs[θs]}, {θs, -180, θ2e, .1}];

In[55]:= ListPlot[points,
  PlotStyle → Gray, PlotRange → {{0, 360}, {-60, 60}},
  Epilog → {Black, Text[Style[#, 32, Bold], #2] & @@@
    {{"◊", {λ1, φ1}}, {"◊", {λ2, φ2}}, {"□", {λ2e, φ2e}}},
  Prolog → {Arrow[{{333, 33}, {342, 31}}, Arrow[{{212, 29}, {215, 30}}],
    Text[Style["Burnout\nφ1, λ1", 10], {280, 18}],
    Text[Style["Selected\nposition\nφ2, λ2", 10], {240, 48}],
    Text[Style["Equivalent\nposition\nφ2, λ2 e", 10], {312, 46}]},
  GridLines → {Table[λ, {λ, 0, 360, 40}], Table[φ, {φ, -60, 60, 20}]},
  Axes → False, Frame → True, ImageSize → 600, AspectRatio → 0.7,
  FrameLabel → {"Longitude, λ, deg", "Latitude, φ, deg"},
  FrameTicks → {{Table[{φ, φ}, {φ, -60, 60, 20}], None}, {Table[{λ, λ}, {λ, 0, 360, 40}], None}}]
```



For comparison, here is her original diagram:



```
In[56]:= path = {#1, QuantityMagnitude[GeodesyData["Krassovsky",
{"FromGeocentricLatitude", #2}]]] & @@@ Flatten[points, 1];
```

A more visually useful version can be constructed using [GeoGraphics](#), taking care to convert the geocentric coordinates into geodetic coordinates:

```
In[56]:= path = {#1, QuantityMagnitude[GeodesyData["Krassovsky",
{"FromGeocentricLatitude", #2}]]] & @@@ Flatten[points, 1];
```

```
In[57]:= GeoGraphics[{Blue, PointSize[0.015], Point[GeoPosition[{φ1, λ1}]], Red,
Point[GeoPosition[{φ2, λ2}]],
Black, PointSize[0.002], Point[GeoPosition[Reverse /@ path]], GeoRange → "World",
GeoZoomLevel → 3]
```



How to Calculate Orbits Today

Today, virtually every one of us has, within immediate reach, access to computational resources far more powerful than those available to the entirety of NASA in the 1960s. Now, using only a desktop computer and the Wolfram Language, you can easily find direct numerical solutions to problems of orbital mechanics such as those posed to Katherine Johnson and her team. While perhaps less taxing of our ingenuity than older methods, the results one can get from these explorations are no less interesting or useful.

To solve for the azimuthal angle ψ using more modern methods, let's set up parameters for a simple circular orbit beginning after burnout over Florida, assuming a spherically symmetric Earth (I'll not bother trying to match the orbit of the Johnson paper precisely, and I'll redefine certain quantities from above using the modern SI system of units). Starting from the same low-Earth orbit altitude used by Johnson, and using a little spherical trigonometry, it is straightforward to derive the initial conditions for our orbit:



```

v0 = Sqrt[G M / r0] (* initial velocity of the circular orbit *);
R = RotationTransform[λ1, {0, 0, 1}] (* used to rotate initial conditions *);
{x0, y0, z0} = R[{r0 Cos[φ1], 0, r0 Sin[φ1]}] (* initial x,y,z at burnout point *);
{vx0, vy0, vz0} = R[v0 {-Cos[φ1] Sin[ψ] - Cos[ψ] Cos[φ1] Sin[φ1],
Cos[ψ] Sin[ψ], Cos[ψ] Cos[φ1] Cos[ψ] - Sin[ψ] Sin[φ1]}] (* initial velocity at the burnout point *)

```

```
ln[65]:= initCond[ψ_, γ_] = {x[0] == x0, y[0] == y0, z[0] == z0,  
                                x'[0] == vx0, y'[0] == vy0, z'[0] == vz0};
```

The relevant physical parameters can be obtained directly from within the Wolfram Language:

```
In[66]:= {M, R, G} = QuantityMagnitude[Flatten[{Entity["Planet",  
"Earth"]][{"Mass", "Radius"}], Quantity[1, "GravitationalConstant"]], "SIBase"];
```

Next, I obtain a differential equation for the motion of our spacecraft, given the gravitational field of the Earth. There are several ways you can model the gravitational potential near the Earth. Assuming a spherically symmetric planet and utilizing a Cartesian coordinate system throughout, the potential is merely:

```
In[67]:= pot[{x_, y_, z_}] := -G M / Sqrt[x^2 + y^2 + z^2];
```

Alternatively, you can use a more realistic model of Earth's gravity, where the planet's shape is taken to be an oblate ellipsoid of revolution. The exact form of the potential from such an ellipsoid (assuming constant mass-density over ellipsoidal shells), though complicated (containing multiple elliptic integrals), is available through [EntityValue](#).

```
ln(68) evg = EntityValue[ massive triaxial ellipsoid (physical system) , "GravitationalPotential"];
```

For a general homogeneous triaxial ellipsoid, the potential contains piecewise functions:

```
In[69]:= (pw = Cases[evg, __Piecewise,  $\infty$ ][[1]]) // TraditionalForm // Style[#, 6] &
```

[illegible]

Here, κ is the largest root of $x^2/(a^2+\kappa)+y^2/(b^2+\kappa)+z^2/(c^2+\kappa)=1$. In the case of an oblate ellipsoid, the previous formula can be simplified to contain only elementary functions...

```
In[70]:= Limit[pw[[1, -1]], b -> a] // FullSimplify
```



$$\begin{aligned} & \sqrt{(k+a)(k+c)} \left(\left(-\text{ArcSin}\left[\frac{\sqrt{a^2-c^2}}{\sqrt{k+a^2}}\right] \sqrt{k+a} \sqrt{\frac{a^2-c^2}{k+a^2}} + \sqrt{a^2-c^2} \sqrt{\frac{(k+a^2)^2}{(k+c)^2}} \right)^x - \right. \\ & \left. \left(\text{ArcSin}\left[\frac{\sqrt{a^2-c^2}}{\sqrt{k+a^2}}\right] \sqrt{k+a^2} \sqrt{\frac{a^2-c^2}{k+a^2}} + \sqrt{a^2-c^2} \sqrt{\frac{(a^2-c^2)(k+c^2)}{(k+a^2)^2}} \right)^y + \right. \\ & \left. 2 \text{ArcSin}\left[\frac{\sqrt{a^2-c^2}}{\sqrt{k+a^2}}\right] \sqrt{k+a^2} \sqrt{\frac{a^2-c^2}{k+a^2}} z^2 \right) + \\ & 2a^2 \left(\text{ArcSin}\left[\frac{\sqrt{a^2-c^2}}{\sqrt{k+a^2}}\right] \sqrt{k+a^2} \sqrt{\frac{a^2-c^2}{k+a^2}} \sqrt{(k+a^2)^2(k+c^2)} + (k+c^2)y^2 + (-k+c^2)z^2 \right) / \\ & \left(2(a^2-c^2)^2 \sqrt{(k+a^2)^2(k+c^2)} \right) \end{aligned}$$

... where $k = ((2z^2(a^2-c^2+x^2+y^2) + (-a^2+c^2+x^2+y^2)^2 + z^4)^{1/2} - a^2 - c^2 + x^2 + y^2 + z^2)/2$.

A simpler form that is widely used in the geographic and space science community, and that I will use here, is given by the so-called International Gravity Formula (IGF). The IGF takes into account differences from a spherically symmetric potential up to second order in spherical harmonics, and gives numerically indistinguishable results from the exact potential referenced previously. In terms of four measured geodetic parameters, the IGF potential can be defined as follows:

```
In[71]:= {a, b} = GeodesyData["ITRF00", #] & /@ {"SemimajorAxis", "SemiminorAxis"};
gPole = 9.8321849378 (* g at Earth's pole in m/s^2 *);
gEquator = 9.78903267715 (* g at Earth's equator in m/s^2 *);
With[{k = b gPole / (a gEquator) - 1, e = Sqrt[1 - b^2 / a^2]},
  potIGF[{x_, y_, z_}] :=
    With[{(* latitude *) phi = ArcTan[Sqrt[x^2 + y^2], z]},
      - G M / Sqrt[x^2 + y^2 + z^2] (1 + k Sin[phi]^2) / Sqrt[1 - e^2 Sin[phi]^2]]]
```

I could easily use even better values for the gravitational force through [GeogravityModelData](#). For the starting position, the IGF potential deviates only 0.06% from a high-order approximation:

```
In[75]:= {potIGF[{x0, y0, z0}],
  GeogravityModelData[GeoPosition[GeoPositionXYZ[{x0, y0, z0}]], "Potential"] //
  QuantityMagnitude[#, "SIBase"] &}
```

```
Out[75]:= {-6.04963 × 107, -6.05363 × 107}
```

With these functional forms for the potential, finding the orbital path amounts to taking a gradient of the potential to get the gravitational field vector and then applying Newton's third law. Doing so, I obtain the orbital equations of motion for the two gravity models:


```
In[76]:= grad = -Grad[pot[{x[t], y[t], z[t]}], {x[t], y[t], z[t]}];
gradIGF = -Grad[potIGF[{x[t], y[t], z[t]}], {x[t], y[t], z[t]}];
```



dimensional space. To give these curves context, I will plot them over a texture map of the Earth's surface, projected onto a sphere. Here I construct the desired graphics objects:

```
In[80]:= (*define orbit burnoutDot and burnoutArrow graphics*)
burnoutCoords = {x0, y0, z0};
burnoutDirection = {vx0, vy0, vz0} / v0;
burnoutDot = {Red, Sphere[burnoutCoords, R/30]};
burnoutArrow = {Red, Arrow[{burnoutCoords,
    burnoutCoords + (R/3) burnoutDirection}]};
burnoutLabels[ψ_, γ_] = Graphics3D[{burnoutDot, burnoutArrow};

In[85]:= (*define globe texture and markings*)
earthTexture = Lighter[#, 0.75] &@ ImageReflect[PlanetData["Earth",
    "CylindricalEquidistantTexture"], Bottom];
globe = ParametricPlot3D[R{-Cos[p] Sin[t], -Sin[p] Sin[t], Cos[t]}, {p, 0, 2 Pi}, {t, 0, Pi},
    Mesh -> None, PlotStyle -> Texture[earthTexture],
    TextureCoordinateFunction -> Automatic]
```




```
In[87]:= pole = Graphics3D[{Thick, Red, Line[{{0, 0, -1.3 R}, {0, 0, 1.3 R}}]}];
equator = ParametricPlot3D[R{Cos[t], Sin[t], 0}, {t, 0, 2 Pi}, PlotStyle -> Yellow];
globeMarkings = {pole, equator};
```

While the orbital path computed in an inertial frame forms a periodic closed curve, when you account for the rotation of the Earth, it will cause the spacecraft to pass over different points on the Earth's surface during each subsequent revolution. I can visualize this effect by adding an additional rotation term to the solutions I obtain from `NDSolve`. Taking the number of orbital periods to be three (similar to John Glenn's flight) for visualization purposes, I construct the following [Manipulate](#) to see how the orbital path is affected by the azimuthal launch angle ψ , similar to the study in Johnson's paper. I'll plot both a path assuming a spherical Earth (in white) and another path using the IGF (in green) to get a sense of the size of the oblateness effect (note that the divergence of the two paths increases with each orbit):




```
sol = NDSolve[Join[Fma, initCond[ψs, γ]], {x, y, z}, {t, 0, τ T}];
solIGF = NDSolve[Join[FmaIGF, initCond[ψs, γ]], {x, y, z}, {t, 0, τ T}];
Show[ParametricPlot3D[Evaluate[{ρ Cos[φ - ω t], ρ Sin[φ - ω t],
    z[t]} /. {sol[[1]], solIGF[[1]]}], {t, 0, τ T}, PlotStyle -> {White, Darker[Green]}] /.
    _Line -> Tube[#, 35000],
    globe, globeMarkings, burnoutLabels[ψs, γ],
    Frame -> None, Ticks -> None, PlotRange -> {All, All, {-1.2 R, 1.2 R}},
    RotationAction -> "Clip", ViewPoint -> {0, -2, 0.4}],
    {{ψs, 70 Degree, "start angle"}, -Pi, Pi},
    {{τ, 1.025, "flight time"}, 0.001, 3}, SaveDefinitions -> True]]
```



In the notebook attached to this blog, you can see this `Manipulate` in action, and note the speed at which each new solution is obtained. You would hope that Katherine Johnson and her colleagues at NASA would be impressed!

Now, varying the angle ψ at burnout time, it is straightforward to calculate the position of the spacecraft after, say, three revolutions:

```
In[91]:= posis = Block[{γ = 0.5 Degree, ω = ωCapitale / (360 / (2 Pi) × 60), T = 3.05 × 2 Pi r0 / v0},
    Table[Map[(Most[#] - {0, ω T 360 / (2 Pi)}) &, GeoPosition[GeoPositionXYZ[{x[T], y[T], z[T]}] /.
        NDSolve[Join[FmaIGF, initCond[ψ, γ]], {x, y, z}, {t, 0, T}][[1]]],
        {ψ, 50 Degree, 90 Degree, 1 Degree}];
```



Modeling the Reentry of a Satellite

The movie also mentions Euler's method in connection with the reentry phase. After the initial problem of finding the azimuthal angle has been solved, as done in the previous sections, it's time to come back to Earth. Rockets are fired to slow down the orbiting body, and a complex set of events happens as the craft transitions from the vacuum of space to an atmospheric environment. Changing atmospheric density, rapid deceleration and frictional heating all become important factors that must be taken into account in order to safely return the astronaut to Earth. Height, speed and acceleration as a function of time are all problems that need to be solved. This set of problems can be solved with Euler's method, as done by Katherine Johnson, or by using the differential equation-solving functionality in the Wolfram Language.

For simple differential equations, one can get a detailed step-by-step solution with a specified quadrature method. An equivalent of Newton's famous $F = m a$ for a time-dependent mass $m(t)$ is the so-called ideal rocket equation (in one dimension)...

$$m(t) v'(t) == -v_e m'_p(t),$$

... where $m(t)$ is the rocket mass, v_e the engine exhaust velocity and $m'_p(t)$ the time derivative of the propellant mass. Assuming a constant $m'_p(t)$, the structure of the equation is relatively simple and easily solvable in closed form:

```
In[93]:= DSolve[{(m0 - β t) v'[t] == v_ex β, v[0] == vi}, v[t], t]
```

```
Out[93]= {{v[t] → vi + Log[m0] v_ex - Log[m0 - t β] v_ex}}
```

With initial and final conditions for the mass, I get the celebrated rocket equation ([Tsiolkovsky 1903](#)):

```
In[94]:= FormulaLookup["rocket equation"]
```

```
Out[94]= {RocketEquation}
```

```
In[95]:= FormulaData["RocketEquation"]
```


symbol	description	physical quantity	dimensions
v_f	final speed	Speed	$\{ \{ \text{LengthUnit}, 1 \}, \{ \text{TimeUnit}, -1 \} \}$
m_f	final mass	Mass	$\{ \text{MassUnit}, 1 \}$
m_i	initial mass	Mass	$\{ \text{MassUnit}, 1 \}$
v_e	effective exhaust velocity	Speed	$\{ \{ \text{LengthUnit}, 1 \}, \{ \text{TimeUnit}, -1 \} \}$
v_i	initial speed	Speed	$\{ \{ \text{LengthUnit}, 1 \}, \{ \text{TimeUnit}, -1 \} \}$

In[97]:= **FormulaData@FormulaLookup["rocket equation"][[1]]**

Out[97]= $v_f == \text{Log}\left[\frac{m_i}{m_f}\right] v_e + v_i$

The details of solving this equation with concrete parameter values and e.g. with the classical [Euler method](#) I can get from [Wolfram|Alpha](#). Here are those details together with a detailed comparison with the exact solution, as well as with other numerical integration methods:

In[98]:= **WolframAlpha["use Euler method (2-t)v'(t)=4, v(0)=0, from t=0 to 0.95"]**





```

"SIBase"]; (* mass of Mercury Atlas 6 *)

φ0 = 34.00; λ0 = 241.00 Degree; (* initial latitude and longitude *)

h = QuantityMagnitude[Mercury Atlas 6 (satellite) [average altitude], "Meters"];

In[102]:= X0 = (R + h) {Cos[λ0] Cos[φ0], Sin[λ0] Cos[φ0], Sin[φ0]}; (* position at start of reentry *)
V0 = Sqrt[G M / (R + h)] {-Sin[λ0], Cos[λ0], 0};
(* velocity at start of reentry *)

```

I assume that the braking process uses 1% of the thrust of the stage-one engine and runs, say, for 60 seconds. The equation of motion is:

$$m_{\text{caps}} \mathbf{v}'(t) = \mathbf{F}_{\text{grav}}(\mathbf{x}(t)) + \mathbf{F}_{\text{exhaust}}(t) + \mathbf{F}_{\text{friction}}(\mathbf{x}(t), \mathbf{v}(t))$$

Here, \mathbf{F}_{grav} is the gravitational force, $\mathbf{F}_{\text{exhaust}}(t)$ the explicitly time-dependent engine force and $\mathbf{F}_{\text{friction}}(\mathbf{x}(t), \mathbf{v}(t))$ the friction force. The latter depends via the air density explicitly on the position $\mathbf{x}(t)$ and via the friction law on $\mathbf{v}(t)$.

For the height-dependent air density, I can conveniently use the [StandardAtmosphereData](#) function. I also account for a height-dependent area because of the parachute that opened about 8.5 km above ground:

```

In[104]:= Cd = 1 (* drag coefficient *);
airDensity[X_?Real, _] :=
  QuantityMagnitude[StandardAtmosphereData[Quantity[Norm[X] - R, "Meters"], "Density"],
    "SIBase"]
airFriction[X_List, V_List] := -1/2 area[Norm[X - R]] V.V Cd airDensity[X] V / Sqrt[V.V]
brake[V_List, t_, {α_, T_}] := -If[t < T, α Normalize[V], 0]

```

This gives the following set of coupled nonlinear differential equations to be solved. The last [WhenEvent](#) [. . .] specifies to end the integration when the capsule reaches the surface of the Earth. I use vector-valued position and velocity variables X and V:

```

In[107]:= area[height_] := If[h > 8500, 10, (* parachute *) 2000];

```

With these definitions for the weight, exhaust and air friction force terms...

```

In[108]:= Fgrav[X_, V_, t_] := -G mCapsule M X / Sqrt[X.X]^3
Fexhaust[X_, V_, t_] := brake[V, t, {3000, 60}]
Fairfraction[X_, V_, t_] := airFriction[X, V]

```

... total force can be found via:

```

In[111]:= Ftotal[X_, V_, t_] := Fgrav[X, V, t] + Fexhaust[X, V, t] + Fairfraction[X, V, t]

```

```

In[112]:= odeSystem = {X'[t] == V[t],
  mCapsule V'[t] == Ftotal[X[t], V[t], t],
  WhenEvent[Norm[X[t]] - R == 0, "StopIntegration"]};

```



In this simple model, I neglected the Earth's rotation, intrinsic rotations of the capsule, active flight angle changes, supersonic effects on the friction force and more. The explicit form of the differential equations in coordinate components is the following. The equations that Katherine Johnson solved would have been quite similar to these:



$$\begin{aligned} & \left\{ -\text{If}[t < 60, 3000 \text{ Normalize}[\{vx[t], vy[t], vz[t]\}], 0] - 5 \text{ airDensity}[\{x[t], y[t], z[t]\}] vx[t] \right. \\ & \quad \left. \sqrt{vx[t]^2 + vy[t]^2 + vz[t]^2} - \frac{5.389 \times 10^{17} x[t]}{(x[t]^2 + y[t]^2 + z[t]^2)^{3/2}}, -\text{If}[t < 60, 3000 \text{ Normalize}[\{vx[t], vy[t], vz[t]\}], 0] - \right. \\ \text{Out}[113] = & \quad \left. 5 \text{ airDensity}[\{x[t], y[t], z[t]\}] vy[t] \sqrt{vx[t]^2 + vy[t]^2 + vz[t]^2} - \frac{5.389 \times 10^{17} y[t]}{(x[t]^2 + y[t]^2 + z[t]^2)^{3/2}}, \right. \\ & \quad \left. -\text{If}[t < 60, 3000 \text{ Normalize}[\{vx[t], vy[t], vz[t]\}], 0] - \right. \\ & \quad \left. 5 \text{ airDensity}[\{x[t], y[t], z[t]\}] vz[t] \sqrt{vx[t]^2 + vy[t]^2 + vz[t]^2} - \frac{5.389 \times 10^{17} z[t]}{(x[t]^2 + y[t]^2 + z[t]^2)^{3/2}} \right\} \\ & \text{WhenEvent}\left[-6.3710088 \times 10^6 + \sqrt{\text{Abs}[x[t]]^2 + \text{Abs}[y[t]]^2 + \text{Abs}[z[t]]^2} == 0, \text{StopIntegration}\right] \end{aligned}$$

Supplemented by the initial position and velocity, it is straightforward to solve this system of equations numerically. Today, this is just a simple call to `NDSolve`. I don't have to worry about the method to use, step size control, error control and more because the Wolfram Language automatically chooses values that guarantee meaningful results:


```
In[114]:= tMax = 5000;
inits = {X[0] == X0, V[0] == V0};
nds = NDSolve[Join[odeSystem, inits], {X, V}, {t, 0, 60, tMax}]
```

```
Out[116]= {{X -> InterpolatingFunction[ Domain: {{0., 1440.}} Output dimensions: {3}],
V -> InterpolatingFunction[ Domain: {{0., 1440.}} Output dimensions: {3}]]}}
```

Here is a plot of the height, speed and acceleration as a function of time:

```
In[117]:= plotXVAT[nds_] :=
With[{T = nds[[1, 1, 2, 1, 2]]/60, opts = Sequence[ImageSize -> 220, PlotRange -> All]},
GraphicsRow[{
Plot[(Norm[X[60 t]] - R)/1000 /. nds[[1]], {t, 0, T}, opts,
AxesLabel -> {"time (min)", "height (km)"},
Plot[Norm[V[60 t]]/1000 /. nds[[1]], {t, 0, T}, opts,
AxesLabel -> {"time (min)", "speed (km/s)"},
Plot[Norm[Ftotal[X[60 t], V[60 t], 60 t]]/(9.81 mCapsule) /. nds[[1]], {t, 0, T}, opts,
AxesLabel -> {"time (min)", "acceleration (g)"}],
Spacings -> 0]]
```


```
In[118]:= plotXVAT[nds]
```



Plotting as a function of height instead of time shows that the exponential increase of air density is responsible for the high deceleration. This is not due to the parachute,




```
Evaluate[{{Norm[X[60 t]] - R} / 1000, Norm[Ftotal[X[60 t], V[60 t], 60 t]} / (9.81 mCapsule)} /. nds[[1]],
{t, 0, T}, opts, AxesLabel -> {"height (km)", "acceleration (g)"}, AspectRatio -> 0.6]
```



And here is a plot of the vertical and tangential speed of the capsule in the reentry process:

```
In[120]:= With[{T = nds[[1, 1, 2, 1, 1, 2]], opts = Sequence[ImageSize -> 300, PlotRange -> All],
{Plot[V[t].Normalize[X[t]] / 1000 /. nds[[1]], {t, 0, T}, opts,
AxesLabel -> {"time (s)", "vertical speed (km/s)"},
Plot[(Norm[V[t] - V[t].Normalize[X[t]]] Normalize[V[t]]) / 1000 /. nds[[1]], {t, 0, T}, opts,
AxesLabel -> {"time (s)", "tangential speed (km/s)"}]}]
```




Now I repeat the numerical solution with a fixed-step Euler method:

```
In[121]:= ndsEuler = NDSolve[Join[odeSystem, inits], {X, V}, {t, 0, tMax},
Method -> {"FixedStep", Method -> "ExplicitEuler"}, StartingStepSize -> 0.05]
```


```
Out[121]= {{X -> InterpolatingFunction[{{0., 1.44 x 10^3}}, {3}],
V -> InterpolatingFunction[{{0., 1.44 x 10^3}}, {3}]}}
```

Qualitatively, the solution looks the same as the previous one:



For the used step size of the time integration, the accumulated error is on the order of a few percent. Smaller step sizes would reduce the error (see the previous Wolfram|Alpha output):

```
In[123]:= With[{T = nds[[1, 1, 2, 1, 1, 2]]},
  Plot[100 ((Norm[X[t]] - R /. nds[[1]]) /
    (Norm[X[t]] - R /. ndsEuler[[1]] - 1), {t, 0, T},
    AxesLabel -> {"time (s)", "height error (%)"}]]
```




Note that the landing time predicted by the Euler method deviates only 0.11% from the previous time. (For comparison, if I were to solve the equation with two modern methods, say "BDF" vs. "Adams", the error would be smaller by a few orders of magnitude.)

Now, the reentry process generates a lot of heat. This is where the heat shield is needed. At which height is the most heat per area q generated? Without a detailed derivation, I can, from purely dimensional grounds, conjecture $\dot{q} \sim \rho V^3$:

```
In[124]:= DimensionalCombinations[{"Speed", "MassDensity", "HeatFlux"}]
```

```
Out[124]= {MassDensity Speed^3}
```

```
In[125]:= With[{T = nds[[1, 1, 2, 1, 1, 2]]},
  ParametricPlot[Evaluate[{{(Norm[X[t]] - R) / 1000, airDensity[X[t]] Norm[V[t]]^3} /. nds[[1]]},
    {t, 0, T}, PlotRange -> All, Ticks -> {True, False},
    AxesLabel -> {"height (km)", "heat generated (a.u.)"}, AspectRatio -> 0.6]]
```



Many more interesting things could be calculated ([Hicks 2009](#)), but just like the movie had to fit everything into two hours and seven minutes, I will now end my blog for the sake of time. I hope I can be pardoned for the statement that, with the Wolfram Language, the sky's the limit.

To download this post as a Computable Document Format (CDF) file, [click here](#). [New to CDF?](#) Get your copy for free with this [one-time download](#).

Posted in: [* Best of Blog *](#), [History](#), [Mathematics](#)

[◀ Newer](#)

[How to Use Your Smartphone for Vibration Analysis, Part 1: The Wolfram Language](#)

[Older ▶](#)

[How Many Animals and Arp-imals Can One Find in a Random 3D Image?](#)

RELATED POSTS

[Interning at Wolfram: My Regeneration as a Theoretical Scientist](#)

November 29, 2018

[As of Today, the Fundamental Constants of Physics \(\$c\$, \$h\$, \$e\$, \$k\$, \$N_A\$ \) Are Finally... Constant!](#)

November 16, 2018

[Revisiting the Disputed *Federalist Papers*: Historical Forensics with the Chaos Game Representation and AI](#)

October 11, 2018

6 Comments



The cdf link is broken!

Posted by Andrew Jaffe February 24, 2017 at 5:24 pm

[Reply](#)



Sorry about that Andrew! It should be back working just fine now.

Posted by [Wolfram Blog](#) February 27, 2017 at 11:03 am

[Reply](#)



The download link seems not working.

Posted by Aidong Chen February 25, 2017 at 12:04 am

[Reply](#)



Thanks Jeffrey, Paco, and Michael for a most appropriate blog. I saw this film in Australia several weeks ago, and four Most mathematical scientists would agree that there are only a few good films in which mathematics and mathematicians play a role. This is one such, and I have to say the title is a clever one. Perhaps you guys could respond with a (short?) list of "mathematics" films, not counting the TV series "Numbers", of course. :-)

Cheers

Barrie

Posted by Barrie Stokes February 26, 2017 at 6:13 pm

[Reply](#)



Whew! I'm sure glad this was "straightforward" and not complicated like I feared.

Posted by dj February 27, 2017 at 12:41 pm

[Reply](#)



I'm excited to uncover this site. I need to thank you for some time for this fantastic read !! I definitely loved every part of it. I have saved it as a favorite to see new stuff in your blog.

Posted by [Career Convey](#) November 11, 2018 at 9:42 am

[Reply](#)

Leave a comment

Login with Facebook



Sign in with Twitter

Sign in with LinkedIn

Or **continue as a guest** (your comment will be held for moderation):

Name

[Products & Services](#)[Technologies](#)[Solutions](#)[Support & Learning](#)[Cc](#)

Wolfram



Wolfram|Alpha



BROWSE BY TOPIC

- [* Best of Blog * \(38\)](#)
- [Astronomy \(20\)](#)
- [Books \(7\)](#)
- [Computational Thinking \(18\)](#)
- [Current Events \(7\)](#)
- [Data Analysis and Visualization \(92\)](#)
- [Data Repository \(7\)](#)
- [Design \(23\)](#)
- [Developer Insights \(34\)](#)
- [Digital Humanities \(2\)](#)
- [Education \(104\)](#)
- [Events \(17\)](#)
- [Finance \(16\)](#)
- [Geosciences \(11\)](#)
- [High-Performance Computing \(11\)](#)
- [History \(7\)](#)
- [Image Processing \(41\)](#)
- [Machine Learning \(7\)](#)
- [Mathematica News \(54\)](#)
- [Mathematica Q&A \(13\)](#)
- [Mathematics \(93\)](#)
- [New Technology \(6\)](#)
- [Other Application Areas \(52\)](#)
- [Raspberry Pi \(16\)](#)
- [Recreational Computation \(109\)](#)
- [Software Development \(24\)](#)
- [SystemModeler \(38\)](#)
- [Wolfram Cloud \(19\)](#)
- [Wolfram Community \(7\)](#)
- [Wolfram Demonstrations Project \(31\)](#)
- [Wolfram Language \(141\)](#)
- [Wolfram News \(209\)](#)
- [Wolfram|Alpha \(26\)](#)
- [Wolfram|One \(4\)](#)





[WolframAlpha.com](#) | [WolframCloud.com](#)

-
- [Products & Services](#)
 - [Technologies](#)
 - [Solutions](#)
 - [Support & Learning](#)
 - [Cc](#)